
Subject: Re: GENERAL SMOKING PREVALENCE IN RWANDA
Posted by Mlue on Mon, 20 Feb 2017 23:52:30 GMT
View Forum Message <> Reply to Message

Hi, I figured out how to compute "Other tobacco"

See Stata code below using Rwanda 2014-15 as an example...

** Rwanda: Standard DHS, 2014-15

// WOMEN'S FILE

********************************************************************************

*** YEAR = 2014-15
** PREVALENCE OF SMOKING

********************************************************************************
** WEIGHT VARIABLE
gen weight = v005/1000000

********************************************************************************

** SURVEY SET
gen psu =    v021
gen strata = v022

svyset psu [pw = weight], strata(strata) vce(linearized)

*******************

** SVYDESCRIBE 
/* This checks whether or not your svyset agrees with what has been published 
on the country's DHS report */

//svydescribe

********************************************************************************

// RENAME

rename v013 age
rename v106 education
rename v190 wealth
rename v025 residence
rename v024 province
rename sdistrict district

*=========================================================================
=====*
// EDUCATION RECODE
recode education (0=0 "No education") (1=1 "Primary") ///
(2/3=2 "Secondary/higher"), gen(education_recode)
label var education_recode "Recode for education"
label val education_recode education_recode

*=========================================================================
=====*

// SMOKING INDICATORS //

** CIGARETTES
recode v463a (. 0 9 = 0 "No") (1 = 1 "Yes"), gen(cigarettes)
label var cigarettes "Smokes cigarettes"
label val cigarettes cigarettes

*=========================================================================
=====*

** PIPE
recode v463b (0 9 = 0 "No") (1 = 1 "Yes"), gen(pipe)
label var pipe "Smokes pipe"
label val pipe pipe

*=========================================================================
=====*

** OTHER TOBACCO

gen other_tobacco	=	0
label define other_tobacco 0"No" 1"Yes"
label var other_tobacco "Uses/smokes other tobacco"
label val other_tobacco other_tobacco

** OTHER TOBACCO

foreach xvar of varlist 463c 463d 463x {
replace other_tobacco=1 if `xvar'==1
}

*=========================================================================
=====*

** DOES NOT USE TOBACCO
recode v463z (0 9 =0) (1=1 "Not smoking"), gen(does_not_use_tobacco)

label var does_not_use_tobacco "Does not use tobacco products"
label val does_not_use_tobacco does_not_use_tobacco

*=========================================================================
=====*

** DROP IF NOT WITHIN SAMPLE
qui regr cigarettes pipe other_tobacco does_not_use_tobacco  [pw=weight]		
drop if e(sample)!=1		/* drop observations with missings on any variable to be used in analysis */

********************************************************************************
** CHECK
svy: tab cigarettes, count format(%4.0f)
svy: tab cigarettes, percent format(%4.1f)

svy: tab pipe, count format(%4.0f)
svy: tab pipe, percent format(%4.1f)

svy: tab other_tobacco, count format(%4.0f)
svy: tab other_tobacco, percent format(%4.1f) 

svy: tab does_not_use_tobacco, count format(%4.0f)
svy: tab does_not_use_tobacco, percent format(%4.1f)

********************************************************************************

